Black Holes and Instantons. Exercise sheet $n^o 5$

Remarks and conventions

This notebook is adapted from the one downloadable on the webpage

http://web.physics.ucsb.edu/~gravitybook/mathematica.html

(select “Curvature and the Einstein Equation”).

The definitions of the various quantities are given by the following formulas. The basic input is the metric

$g_{\alpha\beta}$

Its the inverse is denoted

$g^{\lambda\sigma}$.

Then we define:
the Christoffel symbols or affine connection,

$$\Gamma^{\lambda}{}_{\mu\nu} = \frac{1}{2} g^{\lambda\sigma}(\partial_\mu g_{\sigma\nu} + \partial_\nu g_{\sigma\mu} - \partial_\sigma g_{\mu\nu}),$$

(∂_α stands for the partial derivative $\partial/\partial x^\alpha$),
the Riemann tensor,

$$R^{\lambda}{}_{\mu\nu\sigma} = \partial_\nu \Gamma^{\lambda}{}_{\mu\sigma} - \partial_\sigma \Gamma^{\lambda}{}_{\mu\nu} + \Gamma^{\eta}{}_{\mu\sigma} \Gamma^{\lambda}{}_{\eta\nu} - \Gamma^{\eta}{}_{\mu\nu} \Gamma^{\lambda}{}_{\eta\sigma},$$

the Ricci tensor

$$R_{\mu\nu} = R^{\lambda}{}_{\mu\lambda\nu},$$

the scalar curvature,

$$R = g^{\mu\nu} R_{\mu\nu},$$

and the Einstein tensor,

$$G_{\mu\nu} = R_{\mu\nu} - \frac{1}{2} g_{\mu\nu} R.$$

Preliminaries

In this section we execute some basic command in order for the rest of the notebook to work.

Clearing the values of symbols:

```
Clear[coord, metric, inversemetric,
 affine, riemann, ricci, scalar, einstein, v, r, θ, ϕ]
```

Setting the dimension:

```
n = 4
```

```
4
```

Defining a list of coordinates:

Since we will do all computations in this notebook using the same coordinates (v,r,θ,ϕ), we can already define them:

```
coord = {v, r, θ, ϕ}
```

```
{v, r, θ, ϕ}
```

(a) Einstein ' s equation for the Schwarzschild metric

In the absence of a cosmological constant and matter, the Einstein's equations are simply

$R_{\mu\nu}$=0.

Let us compute $R_{\mu\nu}$.

Defining the metric:

```
metric =
 {{-(1 - 2 GM / r), 1, 0, 0}, {1, 0, 0, 0}, {0, 0, r^2, 0}, {0, 0, 0, r^2 Sin[θ]^2}}
```

$$\left\{\left\{-1+\frac{2\,\mathrm{GM}}{r}, 1, 0, 0\right\}, \{1, 0, 0, 0\}, \left\{0, 0, r^2, 0\right\}, \left\{0, 0, 0, r^2\,\mathrm{Sin}[\theta]^2\right\}\right\}$$

To control the above input, it is useful to display it as a matrix:

```
metric // MatrixForm
```

$$\begin{pmatrix} -1+\frac{2\,\mathrm{GM}}{r} & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2\,\mathrm{Sin}[\theta]^2 \end{pmatrix}$$

Calculating the inverse metric:

```
inversemetric = Simplify[Inverse[metric]]
```

$$\left\{\{0, 1, 0, 0\}, \left\{1, 1 - \frac{2\,GM}{r}, 0, 0\right\}, \left\{0, 0, \frac{1}{r^2}, 0\right\}, \left\{0, 0, 0, \frac{\text{Csc}[\theta]^2}{r^2}\right\}\right\}$$

Calculating the Christoffel symbols:

The final ";" in this command (and any other) prevent *Mathematica* from displaying the result of the evaluation.

```
affine = Simplify[Table[(1 / 2) * Sum[(inversemetric[[i, s]]) *
        (D[metric[[s, j]], coord[[k]] ] +
          D[metric[[s, k]], coord[[j]] ] - D[metric[[j, k]], coord[[s]] ]),
       {s, 1, n}],
     {i, 1, n}, {j, 1, n}, {k, 1, n}] ];
```

Displaying the Christoffel symbols:

Only the non-zero components are displayed using the following procedure.

```
listaffine := Table[If[UnsameQ[affine[[i, j, k]], 0],
   {ToString[Γ[i, j, k]], affine[[i, j, k]]}] , {i, 1, n}, {j, 1, n}, {k, 1, j}]
```

```
TableForm[Partition[DeleteCases[Flatten[listaffine], Null], 2],
 TableSpacing → {2, 2}]
```

Γ[1, 1, 1]	$\frac{GM}{r^2}$
Γ[1, 3, 3]	$-r$
Γ[1, 4, 4]	$-r\,\text{Sin}[\theta]^2$
Γ[2, 1, 1]	$\frac{GM\,(-2\,GM+r)}{r^3}$
Γ[2, 2, 1]	$-\frac{GM}{r^2}$
Γ[2, 3, 3]	$2\,GM - r$
Γ[2, 4, 4]	$(2\,GM - r)\,\text{Sin}[\theta]^2$
Γ[3, 3, 2]	$\frac{1}{r}$
Γ[3, 4, 4]	$-\text{Cos}[\theta]\,\text{Sin}[\theta]$
Γ[4, 4, 2]	$\frac{1}{r}$
Γ[4, 4, 3]	$\text{Cot}[\theta]$

Calculating and displaying the Riemann tensor:

Only the non-zero components are displayed using the following procedure.

```
riemann = Simplify[Table[
    D[affine[[i, j, l]], coord[[k]] ] - D[affine[[i, j, k]], coord[[l]] ] +
     Sum[affine[[s, j, l]] affine[[i, k, s]] - affine[[s, j, k]] affine[[i, l, s]],
      {s, 1, n}],
    {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}] ];
```

```
listriemann := Table[If[UnsameQ[riemann[[i, j, k, l]], 0],
    {ToString[R[i, j, k, l]], riemann[[i, j, k, l]]}] ,
  {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, k - 1}]
```

```
TableForm[Partition[DeleteCases[Flatten[listriemann], Null], 2],
 TableSpacing → {2, 2}]
```

R[1, 1, 2, 1]	$-\frac{2\,GM}{r^3}$
R[1, 3, 3, 1]	$\frac{GM}{r}$
R[1, 4, 4, 1]	$\frac{GM\,Sin[\theta]^2}{r}$
R[2, 1, 2, 1]	$\frac{2\,GM\,(2\,GM-r)}{r^4}$
R[2, 2, 2, 1]	$\frac{2\,GM}{r^3}$
R[2, 3, 3, 2]	$\frac{GM}{r}$
R[2, 4, 4, 2]	$\frac{GM\,Sin[\theta]^2}{r}$
R[3, 1, 3, 1]	$\frac{GM\,(-2\,GM+r)}{r^4}$
R[3, 1, 3, 2]	$-\frac{GM}{r^3}$
R[3, 2, 3, 1]	$-\frac{GM}{r^3}$
R[3, 4, 4, 3]	$-\frac{2\,GM\,Sin[\theta]^2}{r}$
R[4, 1, 4, 1]	$\frac{GM\,(-2\,GM+r)}{r^4}$
R[4, 1, 4, 2]	$-\frac{GM}{r^3}$
R[4, 2, 4, 1]	$-\frac{GM}{r^3}$
R[4, 3, 4, 3]	$\frac{2\,GM}{r}$

Calculating and displaying the Ricci tensor:

Only the non-zero components are displayed using the following procedure.

```
ricci = Simplify[Table[Sum[riemann[[i, j, i, l]], {i, 1, n}], {j, 1, n}, {l, 1, n}] ];
```

```
listricci := Table[If[UnsameQ[ricci[[j, l]], 0],
    {ToString[R[j, l]], ricci[[j, l]]}] , {j, 1, n}, {l, 1, j}]
```

```
TableForm[Partition[DeleteCases[Flatten[listricci], Null], 2],
 TableSpacing → {2, 2}]
```

```
{}
```

We have thus shown explicitly that the Schwarzschild metric satisfies vacuum Einstein's equations.

(b) Computation of the curvature invariant $R_{\mu\nu\rho\sigma} R^{\mu\nu\rho\sigma}$

The curvature invariant $R_{\mu\nu\rho\sigma} R^{\mu\nu\rho\sigma}$ may be computed using the following command:

```
Simplify[Sum[riemann[[i1, j1, k1, l1]] metric[[i1, i2]] inversemetric[[j1, j2]]
   inversemetric[[k1, k2]] inversemetric[[l1, l2]] riemann[[i2, j2, k2, l2]],
  {i1, 1, n}, {j1, 1, n}, {k1, 1, n}, {l1, 1, n}, {i2, 1, n},
  {j2, 1, n}, {k2, 1, n}, {l2, 1, n}]]
```

$$\frac{48\,\text{GM}^2}{r^6}$$

In particular we see that when r->0, this quantity diverge. Since it is a scalar, this is a covariant statement. We have thus explicitly shown that r=0 is a curvature singularity of the Schwarzschild metric.

(c) Computation of the surface gravity κ

The surface gravity κ is defined by the relation

$\xi^\nu \nabla_\nu \xi^\mu = \kappa\, \xi^\mu$ (at the horizon)

where $\xi = \partial_v$. Thus the only non-trivial condition is obtained by focusing on the v-component of the above equality between vectors. We get

$(\nabla_v \xi)^v = \Gamma^v{}_{vv} = \kappa$ (at the horizon)

So all we need to do is to compute $\Gamma^v{}_{vv}$. Since the coordinate v is the first in the list of coordinates, we simply need to extract the [[1,1,1]] component of the table “affine”:

```
affine[[1, 1, 1]]
```

$$\frac{\text{GM}}{r^2}$$

We now evaluate this on the horizon which is, in our coordinates, at r=2GM:

```
% /. r → 2 GM
```

$$\frac{1}{4\,\text{GM}}$$

Thus for the Schwarzschild metric, the surface gravity is κ=1/(4GM); in particular, it is constant along the horizon.

(d) The Vaidya metric

We now proceed to the study of the Vaidya metric. The only difference with the Schwarzschild metric is that the mass is replaced by a function of v, that we call M(v). This is easily implemented in *Mathematica*: we define

```
metric = {{-(1 - 2 G M[v] / r), 1, 0, 0},
  {1, 0, 0, 0}, {0, 0, r^2, 0}, {0, 0, 0, r^2 Sin[θ]^2}}
```

$$\left\{\left\{-1+\frac{2\,\text{G}\,\text{M}[\text{v}]}{r},\ 1,\ 0,\ 0\right\},\ \{1,\ 0,\ 0,\ 0\},\ \left\{0,\ 0,\ r^2,\ 0\right\},\ \left\{0,\ 0,\ 0,\ r^2\,\text{Sin}[\theta]^2\right\}\right\}$$

Since we wrote M[v], *Mathematica* will automatically consider it as a function; in particular, when computing curvature tensors (see below), derivatives of M[v] will appear, as it should.

```
metric // MatrixForm
```

$$\begin{pmatrix} -1+\frac{2\,G\,M[v]}{r} & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2\,\mathrm{Sin}[\theta]^2 \end{pmatrix}$$

```
inversemetric = Simplify[Inverse[metric]]
```

$$\left\{\{0,1,0,0\},\left\{1,1-\frac{2\,G\,M[v]}{r},0,0\right\},\left\{0,0,\frac{1}{r^2},0\right\},\left\{0,0,0,\frac{\mathrm{Csc}[\theta]^2}{r^2}\right\}\right\}$$

```
inversemetric // MatrixForm
```

$$\begin{pmatrix} 0 & 1 & 0 & 0 \\ 1 & 1-\frac{2\,G\,M[v]}{r} & 0 & 0 \\ 0 & 0 & \frac{1}{r^2} & 0 \\ 0 & 0 & 0 & \frac{\mathrm{Csc}[\theta]^2}{r^2} \end{pmatrix}$$

```
affine = Simplify[Table[(1 / 2) * Sum[(inversemetric[[i, s]]) *
        (D[metric[[s, j]], coord[[k]] ] +
          D[metric[[s, k]], coord[[j]] ] - D[metric[[j, k]], coord[[s]] ]),
      {s, 1, n}],
    {i, 1, n}, {j, 1, n}, {k, 1, n}] ];

listaffine := Table[If[UnsameQ[affine[[i, j, k]], 0],
   {ToString[Γ[i, j, k]], affine[[i, j, k]]}] , {i, 1, n}, {j, 1, n}, {k, 1, j}]

TableForm[Partition[DeleteCases[Flatten[listaffine], Null], 2],
 TableSpacing → {2, 2}]
```

Γ[1, 1, 1]	$\frac{G\,M[v]}{r^2}$
Γ[1, 3, 3]	$-r$
Γ[1, 4, 4]	$-r\,\mathrm{Sin}[\theta]^2$
Γ[2, 1, 1]	$\frac{G\left(r\,M[v]-2\,G\,M[v]^2+r^2\,M'[v]\right)}{r^3}$
Γ[2, 2, 1]	$-\frac{G\,M[v]}{r^2}$
Γ[2, 3, 3]	$-r+2\,G\,M[v]$
Γ[2, 4, 4]	$-(r-2\,G\,M[v])\,\mathrm{Sin}[\theta]^2$
Γ[3, 3, 2]	$\frac{1}{r}$
Γ[3, 4, 4]	$-\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]$
Γ[4, 4, 2]	$\frac{1}{r}$
Γ[4, 4, 3]	$\mathrm{Cot}[\theta]$

```
riemann = Simplify[Table[
    D[affine[[i, j, l]], coord[[k]] ] - D[affine[[i, j, k]], coord[[l]] ] +
     Sum[affine[[s, j, l]] affine[[i, k, s]] - affine[[s, j, k]] affine[[i, l, s]],
      {s, 1, n}],
    {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, n}] ];

listriemann := Table[If[UnsameQ[riemann[[i, j, k, l]], 0],
   {ToString[R[i, j, k, l]], riemann[[i, j, k, l]]}] ,
  {i, 1, n}, {j, 1, n}, {k, 1, n}, {l, 1, k - 1}]
```

```
TableForm[Partition[DeleteCases[Flatten[listriemann], Null], 2],
 TableSpacing → {2, 2}]
```

R[1, 1, 2, 1]	$-\frac{2\,G\,M[v]}{r^3}$
R[1, 3, 3, 1]	$\frac{G\,M[v]}{r}$
R[1, 4, 4, 1]	$\frac{G\,M[v]\,\mathrm{Sin}[\theta]^2}{r}$
R[2, 1, 2, 1]	$\frac{2\,G\,M[v]\,(-r+2\,G\,M[v])}{r^4}$
R[2, 2, 2, 1]	$\frac{2\,G\,M[v]}{r^3}$
R[2, 3, 3, 1]	$-G\,M'[v]$
R[2, 3, 3, 2]	$\frac{G\,M[v]}{r}$
R[2, 4, 4, 1]	$-G\,\mathrm{Sin}[\theta]^2\,M'[v]$
R[2, 4, 4, 2]	$\frac{G\,M[v]\,\mathrm{Sin}[\theta]^2}{r}$
R[3, 1, 3, 1]	$\frac{G\left(r\,M[v]-2\,G\,M[v]^2+r^2\,M'[v]\right)}{r^4}$
R[3, 1, 3, 2]	$-\frac{G\,M[v]}{r^3}$
R[3, 2, 3, 1]	$-\frac{G\,M[v]}{r^3}$
R[3, 4, 4, 3]	$-\frac{2\,G\,M[v]\,\mathrm{Sin}[\theta]^2}{r}$
R[4, 1, 4, 1]	$\frac{G\left(r\,M[v]-2\,G\,M[v]^2+r^2\,M'[v]\right)}{r^4}$
R[4, 1, 4, 2]	$-\frac{G\,M[v]}{r^3}$
R[4, 2, 4, 1]	$-\frac{G\,M[v]}{r^3}$
R[4, 3, 4, 3]	$\frac{2\,G\,M[v]}{r}$

```
ricci = Simplify[Table[Sum[riemann[[i, j, i, l]], {i, 1, n}], {j, 1, n}, {l, 1, n}] ];
```

```
Table[If[UnsameQ[ricci[[j, l]], 0], {ToString[R[j, l]], ricci[[j, l]]}] ,
  {j, 1, n}, {l, 1, j}];
```

```
TableForm[Partition[DeleteCases[Flatten[listricci], Null], 2],
 TableSpacing → {2, 2}]
```

R[1, 1]	$\frac{2\,G\,M'[v]}{r^2}$

```
scalar = Simplify[Sum[inversemetric[[i, j]] ricci[[i, j]], {i, 1, n}, {j, 1, n}] ]
```

0

To determine the energy-momentum tensor $T_{\mu\nu}$ which would source the Vaidya metric, we must compute the Einstein's tensor $G_{\mu\nu}$.

```
einstein = Simplify[ricci - (1 / 2) scalar * metric];
```

```
listeinstein := Table[If[UnsameQ[einstein[[j, l]], 0],
   {ToString[G[j, l]], einstein[[j, l]]}] , {j, 1, n}, {l, 1, j}]
```

```
TableForm[Partition[DeleteCases[Flatten[listeinstein], Null], 2],
 TableSpacing → {2, 2}]
```

G[1, 1]	$\frac{2\,G\,M'[v]}{r^2}$

Since Einstein’s equation in the presence of matter are

$G_{\mu\nu} = 8\,\pi\,G\,T_{\mu\nu}$

we deduce that the energy-momentum tensor must have only one non-zero component $T_{\text{vv}} = M'(v)/(4\,\pi\,r\,\hat{}\,2)$.